
1.0 Executive Summary

This memo describes an ANSYS
 electromagnetic-stress analysis of the NCSX TF and PF Coil Systems. The 3D model is fully developed, and salient structural results for five limiting load cases are presented. 
Preliminary results indicate that when the TF coil is charged to 0.5 T, it produces a Cu stress of about 140 MPa (20 ksi) in the transition (de-wedged) region of the inboard leg. This compares favorably with an allowable stress of 250 MPa (36 ksi). Away from this region, the stresses are below 50 MPa (7 ksi). Deformed shape plots highlight the source of this stress-riser and give reason to believe that a local redesign could reduce the stresses and improve the Safety Factor (SF). They also show that the Outboard legs move about ½” toward the machine axis at this field level.
Four other load cases are also considered, based on peak coil currents. The maximum stress occurs in the CS winding pack at t=0.0 s of the 1.7 T Ohmic Scenario. A smeared stress of 31.5 MPa scales to about 50 MPa, resulting in a Safety Factor of ~3.5. The PF5 & PF6 ring coils are also pushed hardest by this time-point. But with unsmeared stresses below 20 MPa, they also appear to be operating in a reasonably safe design space. The SS coil structure has a maximum stress of 44 MPa appearing in the ring-coil support pads. This is a very modest stress for this material, and represents a safety factor of ~8. 
At 0.206 s of the 320 kA Ohmic Scenario, there is a polarity reversal within the CS (PF1 & PF2 have currents opposed to PF3) which leads to the largest vertical tensile stresses in the un-preloaded CS assembly: +4 MPa. Conversely, at 0.506 s of the same scenario, high currents in PF1 and PF2 produce relatively large compressive stresses. These two operating conditions will likely determine the amount of preload and the flexibility (travel) requirements of the CS preload structure, which is still to be designed.

Note: The winding packs are modeled with relatively coarse elements and may not predict bending and therefore shear stresses with great accuracy. More detailed modeling may be required to qualify insulation shear stresses. 
2.0 Analysis

The stress analysis of the NCSX TF & PF coil systems is based on the finite element ANSYS model shown in Fig. 2.0-1. The coupled electromagnetic-structural model is an accumulation of a number of references:

· The coil support structure (shown in red) is a defeatured version of the PPPL solid model
. 
· Coil geometries (TF, PF, Modular and Plasma) are from a PPPL ANSYS database
.
· Coil currents for various operating scenarios are from a PPPL spreadsheet
.
· [image: image1.jpg]Current Density Contours in TF with 16.2 kA 22§Y45276:)3
Orthotropic Resistivity 17:35:23

(0.1673e-7 parallel to Current, 0.1673 in orthogonal dir.) pdrcoils10
AVG ELEMENT SOLL
STEP=1
SUB =1
TIME=1
Js (AVG)
SMN =.927E+07
SMX =.216E+08
927E+07
.106E+08
.120E+08
J134E+08
.148E+08
161E+08
175E+08
.189E+08
.202E+08
.216E+08

Orthogonal Properties produce non-physical distribution

pdrcoils10 I(TF)=16200A




Smear material properties for PF
 & TF
 winding packs.

Fig. 2.0-1: NCSX Magnet System 120˚ Sector
These inputs are manipulated into a text-based ANSYS batch file (included here as Attachment 4.1) which facilitates changes such as material properties, boundary conditions, coil currents and postprocessing with minimal user-input.

EM Symmetry:

Symmetry in the coil system, particularly the Modular coils, requires modeling 1/3rd of the entire magnet system. The ANSYS user is able to account for the unmodeled coils by telling the program that there is n-fold symmetry about the global Z axis. Here, the command is ESYM,3. 
There are a number of structural constraints which must be imposed so that all of the parts work together and remain constrained in space:

Cyclic-Symmetry:

The nodal displacements (UX, UY, UZ) on each face of the 120˚ wedge must be identical. After forcing an identical mesh on these two faces, this so-called cross-face coupling is achieved by invoking the ANSYS command CPCYC,UX,0.1*k,1,,120,,1 for each of the three displacement degrees of freedom (UX,UY,UZ). Fig. 2.0-2 shows a rather busy representation of the cyclic-symmetry constraints required for this 1/3rd model.
TF Winding-Pack and Support Structure Continuity:

The TF coils are designed to be free-floating in the plane of the “D,” but restrained by the crown-like structures (above and below) from out-of-plane motion. Since the structural slots are oversized compared to the winding pack, a layer of material (like G-10) is added to the model. This is equivalent to inserting shims at assembly. Fig. 2.0-3 shows this filler layer between the winding pack and structure. 

The node points of the TF coil and shim elements do not coincide. So achieving displacement continuity across a discontinuous mesh requires writing constraint equations for the local normal displacement degree of freedom. The ANSYS command is CEINTF,uz, where uz is the locally-oriented displacement degree of freedom normal to the side-faces of the winding pack and structure slot. The application of this command requires that the interface between the two surfaces be in close proximity, and this is why the filler material is added to the model. A pictorial representation of the boundary condition is shown in Fig. 2.0-4.
TF Coil Displacement Constraints:

Although the TF coils are constrained from moving out-of-plane by the crown-like structures, and constrained from moving towards the vertical axis of the machine by toroidal continuity (wedging between adjacent inboard legs), they must also be constrained in the vertical direction. This is illustrated in Fig. 2.0-5.
PF Coil to Structure Constraints:

The PF coils 4 through 6 are “attached” to the support structure by coupling the vertical and toroidal displacements at the various interfaces. As with the TF-Shim interaction, the discontinuous mesh at these interfaces requires the use of Constraint Equations. These equations are represented pictorially by the small triads in the element plot of Fig. 2.0-6.

Coil Support Structure Constraints:
The coil support structure is anchored to the Modular Coil support shell at a number of bolted flanges, as shown in the drawings on the bottom of Fig. 2.0-7. The model simulates this fixity by imposing nodal displacement constraints at these flange surfaces, represented by the yellow triads in the model plot shown on the top of Fig. 3.0-7. The effects of cool-down to LN2 temperature (~80K) is included by assigning a coefficient of thermal expansion (CTE) to the various materials (here, the SS support structure), and  imposing displacements (U) consistent with this condition. In this case, the displacement of each node (i) located at the bolted flange surfaces are assigned a vertical (UZ) and radial (UR) displacement determined by the following:
UZi = (Zi)(CTE)(ΔT)
URi = (Ri)(CTE)(ΔT)

Here, Zi is the distance of node-i from the horizontal equatorial plane and Ri is the radius of node-i. The sign of ΔT is consistent to force the displacement in the proper direction (all UR values are negative, UZ values for nodes above the equator are also negative while UZ values for nodes below the equator are positive).

CS Coil Displacement Constraints:

The details of the CS support are not important to this analysis. However, maintaining numeric stability is important. Fig. 2.0-8 is a plot of the Central Solenoid (CS) assembly composed of PF1, PF2 and PF3, upper and lower pairs. The individual coils are joined together by intercoil shims or spacers, as indicated in the plot. The assembly is held in space by setting vertical (and toroidal) displacements to zero at the indicated anchor point. The coil is free to move radially with thermal contraction and EM loading. 
Since the coil currents are symmetric (for the five load cases postulated here), there should not be any significant reaction force at this location. Of course fault conditions could lead to an asymmetry that the actual support structure will need to address.

Fig. 2.0-2 Cyclic-Symmetry across the 120˚ Sector

Fig. 2.0-3 TF Winding-Pack and Support Structure Lateral Continuity
Fig. 2.0-4 TF Coil-Shim Constraint Equation Symbols

Fig. 2.0-5 TF Coil Vertical Displacement Constraints

Fig. 2.0-6 PF Coil-Structure Constraint Equation Symbols Force Vertical and Toroidal Coupling

Fig. 2.0-7 Coil Support Structure Anchored to Modular Coil Shell at Bolted Flanges

Fig. 2.0-8 CS Coil Displacement Constraints (for numerical stability)

3.0 Analysis

The finite element model described above is used to simulate some of the more limiting Load Cases (LC) based on an array of NCSX operating scenarios [4]. A scan of these scenarios leads to a handful of time-points during which most coils are driven to their maximum current level. They are:
1. 0.5 T TF Only, t=0.0 s (Maximum TF coil current)
2. 1.7 T Ohmic Scenario, t=0.0 s (Maximum ~PF5&6 coil currents)
3. 2.0 T High Beta Scenario, t=0.0 s (Maximum Modular coil currents)
4. 320 kA Ohmic Scenario, t=0.206 s (Maximum PF4 coil current)
5. 320 kA Ohmic Scenario, t=0.506 s (Maximum PF1&2 coil and Plasma currents)
The coil currents associated with each of these select Load Cases (and number of turns per coil, N) are listed in Table 3.0-1 and plotted in Fig. 3.0-1 for visualization purposes. Positive and negative extremes are highlighted in the table by color. An inspection of the tabulated values indicates that these five load cases do not capture all of the peak currents. For example, PF5 is not driven to its extreme value during this subset of time points. LC2 puts the PF5 current at 8284 A, while its maximum current is a mere 1% higher at 8356 A during the 1.7T High Beta Scenario. PF3 has a similar story except the shortfall is 9%. 
Table 3.0-2 summarizes the salient stresses, magnetic force summations and force density (ρf) in the TF, PF4-6, and SS support structure for each of the five reference Load Cases. Detailed magnetic and structural results are presented in the following sections. 

Table 3.0-1 Summary of Currents for Limiting Load Cases
	Load Case>>
	1
	2
	3
	4
	5

	Scenario>>
	0.5 T TF
	1.7T Ohmic
	2T High Beta
	320kA Ohmic
	320kA Ohmic

	Time, s
	0.0
	0.0
	0.0
	0.206
	0.506

	M1 (N=40)
	0
	38141
	40908
	34200
	34200

	M2 (N=40)
	0
	35504
	41561
	32057
	32057

	M3 (N=36)
	0
	35453
	40598
	32184
	32184

	PF1 (N=72)
	0
	-25123
	-15274
	11354
	21858

	PF2 (N=72)
	0
	-25123
	-15274
	11354
	21858

	PF3 (N=72)
	0
	-9698
	-5857
	-11802
	-5975

	PF4 (N=80)
	0
	-7752
	-9362
	-13936
	-9441

	PF5 (N=24)
	0
	8284
	1080
	4563
	4634

	PF6 (N=14)
	0
	-8997
	-24
	5068
	5705

	TF (N=12)
	16200
	-3548
	-1301
	2191
	2191

	Plasma (N=1)
	0
	0
	0
	-320775
	-320775

Table 3.0-2 Magnetic Forces and Smeared Stress Summary, LC1-LC5

	Item
	Load Parameter
	1
	2
	3
	4
	5

	TF
	σvM, MPa
	130
	42.2
	16.7
	26.1
	26.0

	CS
	Fz, kN
	0
	468 Attract
	24 Attract
	67 Attract
	778 Attract

	
	σvM, MPa
	0
	31.5
	12.1
	9.6
	25.9

	
	σZ+, MPa
	0
	1.4
	0.9
	4.0
	3.7

	PF4
	Fz, kN
	0
	182 Attract
	117 Attract
	142 Attract
	39 Repel

	
	σvM, MPa
	0
	5.4
	6.7
	11.8
	4.7

	PF5
	Fz, kN
	0
	215 Repel
	19 Repel
	52 Repel
	43 Repel

	
	σvM, MPa
	0
	9.3
	1.6
	7.0
	7.2

	PF6
	Fz, kN
	0
	149 Attract
	0
	58 Repel
	62 Repel

	
	σvM, MPa
	0
	11.9
	0.4
	6.1
	6.9

	Support

Structure
	σvM, MPa
	0
	44.4
	9.4
	24.2
	26.6


3.1 Load Case 1: 0.5 T TF 

Load Case 1 is not part of the official reference scenarios [4]. However, it represents an important operating condition (TF Only, called B0.5), and will serve as a reference for TF stress as a function of coil current. Stresses at other field levels (Bi) can be scaled by (Bi/B0.5)2. 

Fig. 3.1-1 shows contour plots of the flux density in the TF coils and a small toroidal volume (1 cm square in cross-section) located at a major radius (R0) of 1.4 m. It represents a simple check of the model which employs three-fold magnetic symmetry. The average on-axis field (B0) produced by the TF coils is given by the following:

B0 = μ0NI/2πr
In this Load Case, the number of turns (N) is 18 coils time 12 turns/coil, the conductor current is 16.2 kA, and the radius is 1.4 m. Substituting yields a flux density of 0.4999 T. When the flux density is plotted in just the toroidal volume (lower plot) auto-scale contours range from 0.4980 T to 0.5016 T, with an average value of 0.4998 T. Clearly, the model is implementing the three-fold symmetry and calculating fields properly. (This wasn’t always the case; at one point the symmetry could only be 2n.)

Stresses in the free-floating TF coils are shown in Fig. 3.1-2. Notice that the maximum stress appears as red bands at the top and bottom of the Inboard Legs, where the coils become de-wedged. Away from these high-stress regions, the winding pack operates at stress levels below 50 MPa. The maximum smeared stress value is reported to be 130 MPa (19 ksi). Simplistically unsmearing the stress by a ratio of Cu to Smeared winding pack moduli indicates that the stress level in the Cu conductor is:

σCu ≈ 130 MPa (137/129.5) = 138 MPa (20 ksi)

A preliminary stress evaluation is as follows. The nominal allowable membrane stress (Sm) for the Cu conductor is 24 ksi
, which has a bending stress limit of 1.5Sm or 36 ksi (250 MPa). The high bending stress seen here is due to the rather abrupt change in stiffness which is nicely illustrated in the radial displacement contour plot of Fig. 3.1-3. The plot also indicates that the Outboard Legs move inward (towards the machine axis) by ~1/2” when the coils produce an on-axis field of 0.5 T. 
Fig. 3.1-4 shows plots of the centering forces (upper plot) on the TF Inboard-Equatorial plane elements from this Load Case. These purely radial vectors lead to a fairly uniform toroidal compressive (wedging) stress, as shown in the lower plot. 

The free-floating TF coil does not transmit forces into the support structure when it is the only field source. Fig. 3.1-5 shows the rather trivial stress distribution in the structure for this 0.5T/80K load case, only to demonstrate that the model handles the thermal contraction boundary conditions properly. 

Fig. 3.1-1 Flux Density in TF Coils (upper) and in small toroidal volume at R0 (lower)
[image: image13.jpg]by Constraint Equation:

PF Coil-Structure Continuity
s

Look closely to see vertical and toroidal triads





[image: image14.jpg]Radial Displacements

Toroidal Stiffness of
ed Inboard Legs
its Radial Deflection

Outboard Legs Move I—
Toward Machine Axis 12.2 mm

Loss of Toroidal
Continuity in De-Wedged
Region allows WP to Bend

pdrcoils10 I(TF)=16200A, LC=1

ANSYS 7.1
SEP 18 2003
pdrcoils10
-012202
-010917
-009633
-.008348
-.007064
-005779
-.004495
-00321
-001926
-641E-03

Displ. Units: m



Fig. 3.1-2 Stresses from Energized TF Coils at 0.5 T
Things to notice:

· A relatively coarse mesh results in a smeared stress of 130 MPa (19 ksi)
· Maximum stress at the wedge/de-wedge transition region
· A local re-design of this region could reduce this maximum stress

Radial Displacements from Energized TF Coil (B0=0.5 T, 16.2 kA)

Things to notice:

· Outboard legs move 12.2 mm (1/2”) towards the machine axis

· Abrupt stiffness transition in wedged/de-wedged region produces the bending stresses shown in Fig. 3.1-2

Fig. 3.1-4 EM Force Vectors (upper) and Toroidal Stresses (lower), Inboard-Equatorial Plane Energized TF Coil (B0=0.5 T, 16.2 kA)

Fig. 3.1-5 Stress Distribution in Support Structure (B0=0.5 T, 16.2 kA)
With a maximum stress of 23 Pa, this contour plot indicates that the energized and cold (80K) TF coil system produces essentially no stress in the structure.

3.2 Load Case 2: 1.7T Ohmic Scenario, t=0.0 s

For LC2, PF1&2 and the three Modular coils have relatively high currents. This manifests as relatively high force densities (N/mm3) as shown in Figs. 3.2-1 (PF coil system) and 3.2-2 (Modular Coil System). By contrast, the TF coil current is relatively low, and its force density is substantially lower than that of these other coils (as shown in Fig. 3.2-3). The plot of the Modular coil force density is presented for information only. Structural analysis of these coils is beyond the scope of this memo. 
The smeared stresses in the CS (PF1/2/3) are shown in Fig. 3.2-4. Although the high-stress region occurs in the magnet bore and is obscured from this viewing direction, the legend lists the maximum value at 31.5 MPa. Scaling the smeared stresses to account for the Cu-to-smeared modulus ratio (137GPa/93GPa as is done in sect. 3.1) yields a maximum stress of 46 MPa and a safety factor 3.5 relative to the membrane stress limit of 165 MPa.
The ANSYS plot title also lists the net vertical force acting on the upper and lower coils: -468 kN on the upper set and +468 kN on the lower set. These values have been scaled (by 3) to account for symmetry and so they reflect the force on the entire coil. (Ignore the reference to PF3 in this and any other ANSYS titles of the CS. Referring to the CS as PF3 simplified some coding and allowed looping over coils 3-6 for the CS, PF4, PF5 and PF6.)
The smeared stresses in PF4, PF5 and PF6 are shown in Figs. 3.2-5 through 3.2-7. The ANSYS plot titles also list the net vertical magnetic forces acting on the Upper and Lower coils. Contour values indicate that these coils carry smeared stresses ranging from 5 to 12 MPa. Scaling up the stresses to account for the Cu-to-smeared modulus ratio would only yield stresses on the order of 8 to 18 MPa, which results in a safety factor of 9 relative to the membrane stress limit of 165 MPa.

The smeared stresses in the TF coils are shown in Fig. 3.2-8. Again, the local peak stress appears at the edge of the de-wedged region, reaching 42 MPa. This roughly unsmears to about 45 MPa, and could be compared to the 250 MPa bending stress limit instead of the 165 MPa membrane stress limit. This stress has a safety factor of almost 6.
Stress contours within the SS structure are shown in Fig. 3.2-9. These are equally modest, with a peak value of 44 MPa. This stress is below the allowable stress level of 365 MPa (53 ksi [7]), with a safety factor of 8. The high stress occurs in the region of the PF5 and PF6 supports. At this time-point, the PF5 loads repel, while the PF6U loads attract. More detailed designs will include a strap which holds the coil to its base, and is capable of carrying loads which pull the coil vertically away from the global structure. (Refer back to Table 3.0-2 to see the magnitude of attractive and repelling ring coil forces.)

Fig. 3.2-1 Force Density [N/mm3] in the PF Coils, LC2 (1.7 T Ohmic Scenario, t=0.0 s)

Fig. 3.2-2 Force Density [N/mm3] in the Modular Coils, LC2 (1.7 T Ohmic Scenario, t=0.0 s)

Fig. 3.2-3 Force Density [N/mm3] in the TF Coils, LC2 (1.7 T Ohmic Scenario, t=0.0 s)

Fig. 3.2-4 Smeared Equivalent Stress, CS Coils, LC2 (1.7 T Ohmic Scenario, t=0.0 s)

Intercoil spacers/shims are not shown so that the discrete coil sets are visible.

Fig. 3.2-5 Smeared Equivalent Stress, PF4 Coils, LC2 (1.7 T Ohmic Scenario, t=0.0 s)

Fig. 3.2-6 Smeared Equivalent Stress, PF5 Coils, LC2 (1.7 T Ohmic Scenario, t=0.0 s)

Fig. 3.2-7 Smeared Equivalent Stress, PF6 Coils, LC2 (1.7 T Ohmic Scenario, t=0.0 s)

Fig. 3.2-8 Smeared Equivalent Stress, TF Coils, LC2 (1.7 T Ohmic Scenario, t=0.0 s)

Fig. 3.2-9 Equivalent Stress, Support Structure, LC2 (1.7 T Ohmic Scenario, t=0.0 s)

3.3 Load Case 3: 2T High Beta, t=0.0 s

As shown in the summary Table 3.0-2, the stresses found for LC3 are generally less than those for LC2. The exception is a slightly higher PF4 stress, although the increase is minor. LC4 produces the highest PF4 stresses, and so graphical results for PF4 are shown in the following section.
3.4 Load Cases 4: 320kA Ohmic, t=0.206 s
As indicated in the summary Table 3.0-2, Load Cases 4 produces the largest CS vertical tension stresses and the largest PF4 von Mises stresses. A few plots are included to highlight these results. Fig. 3.4-1 is a plot of the electromagnetic force vectors acting on the elements of the CS. Notice that the opposing polarities of PF3 relative to PF1 & PF2, lead to separating forces. This manifests as a vertical tensile stresses over a substantial fraction of the CS, as shown in the contour plot of Fig. 3.4-2. Clearly, a preload structure is required to offset these tensile stresses and maintain compression in the multi-coil assembly during all operating conditions. The design of such a preload structure is still to be performed. However, this analysis points to one of the more critical load cases in the design of the preload structure.
The stresses in PF4 are a maximum for this LC4. Fig. 3.4-3 shows that the smeared stress is 12 MPa in the coil bore. This unsmears to about 18 MPa, and must be compared to the 165 MPa membrane stress limit. This stress has a safety factor of about 9.

Fig. 3.4-1 EM Force Vectors, Central Solenoid, LC4 (320kA Ohmic, t=0.206 s)

Fig. 3.4-2 Vertical Stresses, Central Solenoid, LC4 (320kA Ohmic, t=0.206 s)

Fig. 3.4-3 Smeared Equivalent Stress, PF4 Coils, LC4 (320kA Ohmic, t=0.206 s)

3.5 Load Cases 5: 320kA Ohmic, t=0.506 s
As indicated in the numerical summary in Table 3.0-2, the only outstanding feature of LC5 are the large net compressive loads developed within the CS. Fig. 3.5-1 is a plot of the vertical stresses in the CS (including Intercoil Shims). With such a large compressive force, this case may be required to define the amount of “travel” required for the CS preload structure.   

Fig. 3.5-1 Vertical Stresses, Central Solenoid, LC4 (320kA Ohmic, t=0.506 s)

4.0 Attachments
4.1 Batch Input File (09/24/03)
/BATCH

rn=15

/filnam,pdrcoils1%rn%

/show,pdrcoils1%rn%,grp

!resume

!*if,1,eq,1,:1000

/PREP7

/com

/com EM-STRUC Analysis of the NCSX PDR TF/PF Coil System

/com

/com

/com  Run History

/com 10: TF Only, 16.2kA, 0.5T, Isotropic resistivity produces best results

/com 11: timepoint=1 (0.5 T TF Only), with CEs on TF sides

/com 12: timepoint=2 (1.7T Ohmic Scenario, t=0.0)

/com 13: timepoint=3 (2T High Beta Scenario, t=0.0)

/com 14: timepoint=4 (320kA Ohmic Scenario, t=0.206)

/com 15: TBD: timepoint=5 (320kA Ohmic Scenario, t=0.506)

/com Set all alpha values to 10e-6

/com 16: timepoint=1 (0.5 T TF Only, 85K), tunif=-215, tref=0, hold PF5/6 by CE

/com 17: timepoint=1 reft=300, bf=85, d,all,temp,85, hold PF5/6 w/CE (Structure stress not right)

/com 18: Just Cold, tref=0, bfunif=-215, d,all,temp,-215, hold PF5/6 w/CE, cnvtol,f,big

/com 19: timepoint=1 tref=0, bfunif=-215, d,all,temp,-215, hold PF4/5/6 w/CE, cnvtol,f,big

/com 11: Just Cold (0.5e-4 T TF), now with CEs on TF sides...looks OK

/com  Now with CEs on TF sides

/com 12: TP=1 (85K and 0.5 T TF)...looks OK

/com 13: TP=2

/com 14: TP=3

/com 15: TP=4

/com 16: TP=5

/com Incorporate the CS into the structural model

/com 111: TP=1

/com 112: TP=2

/com 113: TP=3

/com 114: TP=4

/com 115: TP=5

/com

/com Field Analysis based on coil definitions in HM Fan's file-1.7t.db (saved as pdrcoils1.db)

/com

!resume,pdrcoils1,db,,1      ! pull in PF/TF/MC/Plasma Current Sources and merge parameters

!resume,pdrcoils2,db,,1      ! pull in PF/TF/MC/Plasma Currents (Cheat: PF4 moved 1.12 cm to Struc Surface)

resume,pdrcoils3,db,,1      ! pull in CS/PF/TF/MC/Plasma Currents

/com

/com Misc Parameters

/com

time_pt=5                   ! Coil Currents for this time point (see array listed below)

t=1e-3                      ! tiny

k=0.0254                    ! m/in

emsym,3                     ! number of circular symmetry sections about Global Z

r0=1.4                      ! Major Radius

tmp=-215                    ! Operating Temp

t_ref=0                     ! Reference Temperature

alpha_struc=10e-6           ! Thermal expansion coefficient of Structure

tref,t_ref                 ! Or use MP command

!tunif,tmp                  ! Steve says do not use tunif

save

/pbc,mag

/pbc,temp

/pbc,volt

/pbc,ce

/com

/com Element Types

/com

et,1,5,0                    ! U,TEMP,VOLT,MAG (PF4/5/6 and TF)

et,2,36                     ! Plasma Current Source Primitive

et,3,5,1                    ! TEMP,VOLT,MAG (CS, M and R0 ring)

/com

/com Merge TF Coil nodes in the inboard leg

/com

esel,s,mat,,2

nsle

csys,1

nsel,r,loc,x,,r0

numm,node

/com

/com Material Properties

/com

allsel

mpdele,all,all

/com All PF Coils

!mp,reft,1,t_ref

mp, kxx,1,1

mp,murx,1,1

mp,rsvx,1,1

mp,alpx,1,10e-6    ! roughly (no reference or calc)

/com Chang Jun, "ANSYS Modeling to obtain Equivalent Moduli of Elasticity of 

/com              PF & TF Coils of NCSX," 06/27/03 

mp,  ex,1,62.27E9!,54.44E9

mp,  ey,1,93.10E9!,85.10E9

mp,  ez,1,64.03E9!,56.59E9

mp, Gxy,1,35.27E9!,32.01E9

mp, Gyz,1,35.27E9!,32.01E9

mp, Gxz,1,20.69E9!,18.31E9

mp,nuxy,1,  0.306!,0.310

mp,nuyz,1,  0.213!,0.209

mp,nuxz,1,  0.339!,0.340 

/com TF Coils

!mp,reft,2,t_ref

mp, kxx,2,1

mp,murx,2,1

mp,rsvx,2,1

mp,alpx,2,10e-6    ! roughly (no reference or calc)

/com Chang Jun, "ANSYS Modeling to obtain Equivalent Moduli of Elasticity of 

/com              TF Coils of NCSX," 07/15/03 

mp,  ex,2,111.6E9!, 93.3E9

mp,  ey,2,129.5E9!,119.0E9

mp,  ez,2,125.1E9!,113.9E9

mp, Gxy,2, 48.6E9!, 44.4E9

mp, Gyz,2, 48.6E9!, 44.4E9

mp, Gxz,2, 41.8E9!, 37.5E9

mp,nuxy,2,  0.323!,  0.329

mp,nuyz,2,  0.277!,  0.273

mp,nuxz,2,  0.277!,  0.273

/com M1 Coils

!mp,reft,3,t_ref

mp, kxx,3,1

mp,murx,3,1

mp,rsvx,3,1

/com M2 Coils

!mp,reft,4,t_ref

mp, kxx,4,1

mp,murx,4,1

mp,rsvx,4,1

/com M3 Coils

!mp,reft,5,t_ref

mp, kxx,5,1

mp,murx,5,1

mp,rsvx,5,1

/com R0 Field Elements

!mp,reft,7,t_ref

mp, kxx,7,1

mp,murx,7,1

mp,rsvx,7,1

/com CS Coil-Coil shims (G-10)

!mp,reft,9,t_ref

mp,murx,9,1

mp,  ex,9,25e9

mp,alpx,9,10e-6

mp,nuxy,9,.21

/com Coil Support Structure (Stainless Steel)

!mp,reft,10,t_ref

mp,murx,10,1

mp,  ex,10,205e9

mp,alpx,10,alpha_struc

mp,nuxy,10,0.265

mp, gxy,10,81e9

/com TF Coil-Case side shims (G-10)

!mp,reft,11,t_ref

mp,murx,11,1

mp,  ex,11,25e9

mp,alpx,11,10e-6

mp,nuxy,11,.21

/com 

/com Current Nodes

/com

n_pfcoils=6   ! number of PF Coil Pairs

n_tfcoils=6   ! number of PF Coils

n_mcoils=3    ! number of Modular Coil Pairs

/com PF Coil Current Nodes

n_pf1u=1

n_pf1l=1488

n_pf2u=1110

n_pf2l=1593

n_pf3u=1299

n_pf3l=2346

n_pf4u=190

n_pf4l=1782

n_pf5u=450

n_pf5l=1922

n_pf6u=846

n_pf6l=2126

/com TF Coil Current Nodes

n_tf1=3641

n_tf2=4730

n_tf3=5819

n_tf4=6908

n_tf5=7997

n_tf6=9086

/com Modular Coil Current Nodes

n_m11=13599

n_m12=13609

n_m13=13619

n_m14=15619

n_m21=21639

n_m22=21649

n_m23=21659

n_m24=23659

n_m31=29679

n_m32=29689

n_m33=29699

n_m34=31699

/com Make nodal components out of Modular Coil Current Nodes

*do,jj,1,n_mcoils

nsel,none

*do,j,1,4

nsel,a,node,,n_m%jj%%j%

*enddo

cm,n_m%jj%,node

*enddo

/com

/com Define Coil Currents

/com

*dim,i_coils,array,6,12   ! Time,M1,M2,M3,PF1,PF2,PF3,PF4,PF5,PF6,TF,Plasma

/com Proposed Worst 5 Cases (from Wayne's TDS C08R00 Spreadsheet, circa 09/15/03)

i_coils(1,1)=1,2,3,4,5

i_coils(1,2)=0,38141,40908,34200,34200

i_coils(1,3)=0,35504,41561,32057,32057

i_coils(1,4)=0,35453,40598,32184,32184

i_coils(1,5)=0,-25123,-15274,11354,21858

i_coils(1,6)=0,-25123,-15274,11354,21858

i_coils(1,7)=0,-9698,-5857,-11802,-5975

i_coils(1,8)=0,-7752,-9362,-13936,-9441

i_coils(1,9)=0,8284,1080,4563,4634

i_coils(1,10)=0,-8997,-24,5068,5705

i_coils(1,11)=16200,-3548,-1301,2191,2191

i_coils(1,12)=0,0,0,-320775,-320775

/com

/com Modular Coils

/com Scale currents by tpmc*/2 because each WP has two nodes in the F-command

tpmc1=10*4     ! turns per MC1 winding

tpmc2=10*4     ! turns per MC2 winding

tpmc3=9*4      ! turns per MC3 winding

*do,j,1,n_mcoils

cmsel,s,n_m%j%

f,all,amps,(tpmc%j%/2)*i_coils(time_pt,j+1)

*enddo

/com PF Coils

tppf1=72       ! turns per PF1 winding

tppf2=72       ! turns per PF2 winding

tppf3=72       ! turns per PF3 winding

tppf4=80       ! turns per PF4 winding

tppf5=24       ! turns per PF5 winding

tppf6=14       ! turns per PF6 winding

*do,j,1,n_pfcoils

nsel,s,node,,n_pf%j%u

nsel,a,node,,n_pf%j%l

f,all,amps,(tppf%j%)*i_coils(time_pt,j+4)

*enddo

/com TF Coils

tptf=12

*do,j,1,n_tfcoils

nsel,s,node,,n_tf%j%

f,all,amps,tptf*i_coils(time_pt,11)

*enddo

/com Plasma Current in Type,2 Real,1

r,1,2,i_coils(time_pt,12),0.03,0.05

/com

/com Make a ring of elements on the Plasma Axis

/com

*if,time_pt,eq,1,then

csys

wpcsys

vsel,none

esel,none

nsel,none

cylind,r0-0.005,r0+0.005,-0.005,0.005,-60,60

vatt,7

esize,0.1

type,3

mp,murx,7,1

vmesh,all

d,all,mag

d,all,temp

d,all,volt

*endif

itf=i_coils(time_pt,11)

tme=i_coils(time_pt,1)

/title,pdrcoils1%rn% I(TF)=%itf%A, LC=%time_pt%

allsel

save

fini

/com

/com Pull in Structural Solid

/com

vsel,none

asel,none

lsel,none

ksel,none

/AUX15  

IOPTN,IGES,NODEFEAT 

IOPTN,MERGE,YES 

IOPTN,SOLID,YES 

IOPTN,SMALL,YES 

IOPTN,GTOLER, DEFA  

IGESIN,'inboard_beams_steps_inter_1coiltop_LM10_asm','igs',' '   

fini

/prep7

/com Make structural element

*get,etmx,etyp,,num,max

et,etmx+1,92

et,etmx+2,200,5

/com Mesh Structure

cel=2*k

esize,cel

/com Mesh Inner Volume Faces

csys,1

asel,s,loc,x,,r0

asel,r,loc,z,.5,12

allsel,below,area

*get,thmn_inface,kp,,mnloc,y

*get,thmx_inface,kp,,mxloc,y

asel,r,loc,y,thmn_inface-0.1,thmn_inface+0.1

cm,a1_inface,area

asll,,1

asel,r,loc,y,thmx_inface-0.1,thmx_inface+0.1

cm,a2_inface,area

cmsel,s,a1_inface

type,etmx+2

amesh,all

asll,,1

mshcopy,area,a1_inface,a2_inface,1,,(thmx_inface-thmn_inface),,t

/com Mesh Outer Volume Faces

csys,1

asel,s,loc,x,r0,12

asel,r,loc,z,.5,12

allsel,below,area

*get,thmn_outface,kp,,mnloc,y

*get,thmx_outface,kp,,mxloc,y

asel,r,loc,y,thmn_outface-0.1,thmn_outface+0.1

cm,a1_outface,area

asll,,1

asel,r,loc,y,thmx_outface-0.1,thmx_outface+0.1

cm,a2_outface,area

cmsel,s,a1_outface

type,etmx+2

amesh,all

asll,,1

mshcopy,area,a1_outface,a2_outface,1,,(thmx_outface-thmn_outface),,t

/com Mesh all volumes

allsel

type,etmx+1

vmesh,all

/com Generate the other upper structural set

csys,1

vsel,u,mat,,7

VGEN,2,all,,,,60,,,0

/com Reflect to get the bottom structure

csys

vsymm,z,all

/com Merge Adjacent Nodes

/com Outboard Interface

esel,s,type,,etmx+1

nsle

cm,struc_node,node

nsel,none

cm,merge_node,node

*do,j,1,5

cmsel,s,struc_node

csys

wpcsys

wprot,(j-1)*20-40

cswpla,200

nsel,r,loc,y,-0.10*k,0.10*k

nsel,r,loc,x,r0,12

cmsel,a,merge_node

cm,merge_node,node

*enddo

cmsel,s,merge_node

numm,node,0.050*k*1.25

/com Inboard Interface

esel,s,type,,etmx+1

nsle

csys

nsel,r,loc,y,-0.10*k,0.10*k

nsel,r,loc,x,,r0

numm,node,0.050*k*1.25

nsle

numcmp,node

/com Write Structural Coupling for Cyclic Symmetry

allsel

*get,ncp,cp,,num,max

modmesh,detach

esel,s,type,,1

esel,a,type,,etmx+1

nsle

csys,1

nrotate,all

esel,s,type,,etmx+1

nsle

nsel,r,loc,y,-60.1,-59.9

cm,neg60,node

nsle

nsel,r,loc,y,59.9,60.1

cm,pos60,node

esel,s,mat,,1,2

nsle

nsel,r,loc,y,-60.005,-59.995

cmsel,a,neg60

cm,neg60,node

nsle

nsel,r,loc,y,59.995,60.005

cmsel,a,pos60

cm,pos60,node

/com The cpcyc commands take too long unless the wedge face nodes are the only ones selected

esel,s,mat,,1,2

esel,r,type,,1,3,2

nsle

cmsel,r,pos60

cm,ncoils1,node

esel,s,mat,,1,2

esel,r,type,,1,3,2

nsle

cmsel,r,neg60

cmsel,a,ncoils1

cpcyc,ux,0.1*k,1,,120,,1

cpcyc,uy,0.1*k,1,,120,,1

cpcyc,uz,0.1*k,1,,120,,1

esel,s,mat,,10

esel,r,type,,4

nsle

cmsel,r,pos60

cm,nstruc1,node

esel,s,mat,,10

esel,r,type,,4

nsle

cmsel,r,neg60

cmsel,a,nstruc1

cpcyc,ux,0.1*k,1,,120,,1

cpcyc,uy,0.1*k,1,,120,,1

cpcyc,uz,0.1*k,1,,120,,1

/com

/com Couple UX/UY/UZ at TF outboard equator

/com

esel,s,mat,,2

nsle

csys,1

nsel,r,loc,z

nsel,r,loc,x,r0,12

cpint,ux

cpint,uy

cpint,uz

/com

/com Constrain sides of WP to TF Case Side-Wall Shims

/com

dz_tfcoil=0.135

dz_tfcase=0.153

*do,j,1,6

csys,1

esel,s,mat,,2,11,9

nsle

nsel,r,loc,y,-60+(j-1)*20,-60+(j-0)*20

nsel,u,loc,z,-0.75,0.75

esln,,1

csys,13+8*(j-1)

nsel,u,loc,z,-dz_tfcoil/3,dz_tfcoil/3

nrotate,all

nsle

cm,n_temp,node

cm,e_temp,elem

esel,r,mat,,11

cm,e_keep,elem

esln,,1

esel,r,mat,,2

nsle

nsel,u,loc,z,-dz_tfcoil/4,dz_tfcoil/4

cm,n_keep,node

cmsel,s,n_keep

cmsel,s,e_keep

ceint,,uz

*enddo

/com

/com Constrain midplane of WP (only good for TF On Load Case)

/com

*if,time_pt,eq,1,then

*do,j,1,6

csys,1

esel,s,mat,,2

nsle

csys,13+8*(j-1)

nsel,r,loc,z,-dz_tfcoil/3,dz_tfcoil/3

!nrotate,all

!d,all,uz

*enddo

*endif

/com

/com Couple the Ring Coils to their supports

/com

esel,s,mat,,1

esel,r,type,,1,3,2

nsle

csys,1

nrotate,all

/com CS Radii

esel,s,mat,,9

nsle

csys,1

*get,ro_cs,node,,mxloc,x

*get,ri_cs,node,,mnloc,x

/com PF4

/com

zsup_pf4=1.4478

esel,s,mat,,1

esel,r,type,,1

nsle

csys,1

nsel,r,loc,x,ro_cs+0.001,r0            ! PF4 must be inboard of R0

esln,,1

cm,pf4_elem,elem            ! all PF4 elements

nsel,r,loc,z,,12

esln,,1

cm,pf4u_elem,elem           ! PF4Upper elements

nrotate,all

*get,zbot_pf4,node,,mnloc,z

*get,ro_pf4,node,,mxloc,x

*get,ri_pf4,node,,mnloc,x

esel,invert

cmsel,r,pf4_elem

cm,pf4l_elem,elem           ! PF4Lower elements

nrotate,all

/com Set Up Constraints on PF4U

esel,s,mat,,10

nsle

nsel,r,loc,z,zsup_pf4-t,zsup_pf4+t

nsel,r,loc,x,ri_pf4-cel/4,ro_pf4+cel/4

nrotate,all

cmsel,s,pf4u_elem

ceint,,uz

ceint,,uy

/com Set Up Constraints on PF4L

esel,s,mat,,10

nsle

nsel,r,loc,z,-zsup_pf4-t,-zsup_pf4+t

nsel,r,loc,x,ri_pf4-cel/4,ro_pf4+cel/4

cmsel,s,pf4l_elem

ceint,,uz

ceint,,uy

/com

/com PF5

/com

esel,s,mat,,1

esel,r,type,,1

nsle

csys,1

nsel,u,loc,x,,r0

*get,ri_pf5,node,,mnloc,x

nsel,r,loc,x,ri_pf5-t,ri_pf5+t

nsel,r,loc,z,,12

*get,zbot_pf5,node,,mnloc,z

nsle

nsel,r,loc,x,ri_pf5-t,12

nsel,r,loc,z,zbot_pf5-t,,zbot_pf5+t

*get,ro_pf5,node,,mxloc,x

nsle

nsel,r,loc,x,ri_pf5-t,ro_pf5+t

esln,,1

nrotate,all

cm,pf5_elem,elem

nsel,r,loc,z,,12

esln,,1

cm,pf5u_elem,elem

esel,invert

cmsel,r,pf5_elem

cm,pf5l_elem,elem

/com Set Up Constraints on PF5U

esel,s,mat,,10

nsle

nsel,r,loc,z,zbot_pf5-t,zbot_pf5+t

nsel,r,loc,x,ri_pf5,ro_pf5

nrotate,all

cmsel,s,pf5u_elem

ceint,,uz

ceint,,uy

/com Set Up Constraints on PF5L

esel,s,mat,,10

nsle

nsel,r,loc,z,-zbot_pf5-t,-zbot_pf5+t

nsel,r,loc,x,ri_pf5,ro_pf5

nrotate,all

cmsel,s,pf5l_elem

ceint,,uz

ceint,,uy

/com

/com PF6

/com

esel,s,mat,,1

esel,r,type,,1

cmsel,u,pf4_elem

cmsel,u,pf5_elem

nsle

csys,1

nsel,u,loc,x,,ro_cs+0.001

esln,,1

cm,pf6_elem,elem

nsle

csys,1

nrotate,all

*get,ri_pf6,node,,mnloc,x

*get,ro_pf6,node,,mxloc,x

*get,zbot_pf6,node,,mxloc,z

nsel,r,loc,z,,12

esln,,1

cm,pf6u_elem,elem

esel,invert

cmsel,r,pf6_elem

cm,pf6l_elem,elem

/com Set Up Constraints on PF6U

esel,s,mat,,10

nsle

nsel,r,loc,z,zbot_pf6-t,zbot_pf6+t

nsel,r,loc,x,ri_pf6,ro_pf6

cmsel,s,pf6u_elem

ceint,,uz

ceint,,uy

/com Set Up Constraints on PF6L

esel,s,mat,,10

nsle

nsel,r,loc,z,-zbot_pf6-t,-zbot_pf6+t

nsel,r,loc,x,ri_pf6,ro_pf6

cmsel,s,pf6l_elem

ceint,,uz

ceint,,uy

/com

/com Make a component out of the CS (call it PF3 for logic reasons)

/com

esel,s,type,,1

nsle

csys,1

nsel,r,loc,x,ri_cs-0.001,ro_cs+0.001

esln,,1

cm,cs_elem,elem

cm,pf3_elem,elem

nsel,r,loc,z,,12

esln,,1

cm,pf3u_elem,elem

esel,invert

cmsel,r,pf3_elem

cm,pf3l_elem,elem

/com

/com Restrain The TF Coil in the Global Z

/com

*do,j,1,6

esel,s,mat,,2

nsle

csys,13+8*(j-1)

nsel,r,loc,z

csys,1

*get,zmn2%j%,node,,mnloc,z

nsel,r,loc,z,zmn2%j%-t,zmn2%j%+t

*get,thn2%j%,node,,mnloc,y

*get,r2%j%,node,,mnloc,x

nsel,r,node,,node(r2%j%,thn2%j%,zmn2%j%)

csys,1

nrotate,all

d,all,uz

*enddo

/com

/com Restrain the CS in Z and theta at the bottom

/com

cmsel,s,cs_elem

nsle

*get,zmn_cs,node,,mnloc,z

nsel,r,loc,z,zmn_cs

nsel,r,loc,y

d,all,uy,,,,,uz

/com

/com Restrain the TF Structure in Z & theta (Function of Temperature)

/com

/com At the inner shelf/flange/flats

/com

zin_struc=1.38938   ! Elevation of flat

r1in_struc=0.90     ! outer radius of flat

r2in_struc=1.05     ! inner radius of flat

thin_struc=60       ! angular frequency of supports

dthin_struc=26      ! angular extent of supports

nxdthin_struc=12    ! angular extent of nixed supports

nsel,none

cm,nin_struc,node

*do,j,1,2

esel,s,mat,,10

nsle

csys,1

nsel,r,loc,z,-zin_struc-t,zin_struc+t

nsel,r,loc,x,r1in_struc,r2in_struc

nsel,r,loc,y,(j-1)*thin_struc-30-dthin_struc/2,(j-1)*thin_struc-30+dthin_struc/2

nsel,u,loc,y,(j-1)*thin_struc-30-nxdthin_struc/2,(j-1)*thin_struc-30+nxdthin_struc/2

cmsel,a,nin_struc

cm,nin_struc,node

*enddo

cmsel,s,nin_struc

nrotate,all

d,all,uy!,,,,,uz

*get,nstrt,node,,num,min

*get,nstop,node,,num,max

*do,j,nstrt,nstop,1

*get,nselj,node,j,nsel

*if,nselj,ne,1,cycle

d,j,ux,nx(j)*alpha_struc*(tmp-t_ref)

d,j,uz,nz(j)*alpha_struc*(tmp-t_ref)

*enddo

/com

/com At the outer shelf/flange/flats

/com

zout_struc=1.246691  ! Elevation of flat

r1out_struc=1.98     ! outer radius of flat

r2out_struc=2.05     ! inner radius of flat

thout_struc=20       ! angular frequency of supports

dthout_struc=16      ! angular extent of supports

nxdthout_struc=7.5   ! angular extent of nixed supports

nsel,none

cm,nout_struc,node

*do,j,1,6

esel,s,mat,,10

nsle

csys,1

nsel,r,loc,z,-zout_struc-t,zout_struc+t

nsel,r,loc,x,r1out_struc,r2out_struc

nsel,r,loc,y,(j-1)*thout_struc-50-dthout_struc/2,(j-1)*thout_struc-50+dthout_struc/2

nsel,u,loc,y,(j-1)*thout_struc-50-nxdthout_struc/2,(j-1)*thout_struc-50+nxdthout_struc/2

cmsel,a,nout_struc

cm,nout_struc,node

*enddo

cmsel,s,nout_struc

nrotate,all

d,all,uy!,,,,,uz

*get,nstrt,node,,num,min

*get,nstop,node,,num,max

*do,j,nstrt,nstop,1

*get,nselj,node,j,nsel

*if,nselj,ne,1,cycle

d,j,ux,nx(j)*alpha_struc*(tmp-t_ref)

d,j,uz,nz(j)*alpha_struc*(tmp-t_ref)

*enddo

/com

/com Nix CEs

/com

!esel,all

!nsle

!cedele,all

/com Nix 200 elements 

esel,s,type,,etmx+2

!edele,all

/com

/com Mag BCs

/com

esel,s,type,,1,3,2

nsle

d,all,mag

/com

/com Thermal BCs

/com

/com CS Temp

cmsel,s,cs_elem

nsle

d,all,temp,tmp

/com PF4 Temp

cmsel,s,pf4_elem

nsle

d,all,temp,tmp

/com PF5 Temp

cmsel,s,pf5_elem

nsle

d,all,temp,tmp

/com PF6 Temp

cmsel,s,pf6_elem

nsle

d,all,temp,tmp

/com TF Temp

esel,s,mat,,2

nsle

d,all,temp,tmp

/com Mod Coils

esel,s,mat,,3,5

nsle

d,all,temp,tmp

/com Oh, Just do it to all TEMP nodes

esel,s,type,,1,3,2

nsle

d,all,temp,tmp

/com Structure Temp

allsel

bfunif,temp,tmp

allsel

save

fini

/solu

cnvtol,f,1e8,0.99             ! do not let the program try to converge on F

/com Solve Conduction Problem

allsel

solve

/com Solve for the fields on these nodes...

esel,s,mat,,1,2     ! PFs and TFs

esel,a,mat,,3,5    ! M1/2/3

esel,a,mat,,7       ! R0

esel,r,type,,1,3,2

nsle

/com ...from these elements

esel,s,mat,,1,5

esel,r,type,,1,3

esel,a,type,,2

biot,new

/com Solve for the Stresses

allsel

solve

fini

:1000

/post1

/dev,font,1,Arial,700,0,-19,0,0,,,  

/view,1,1,1,1

/vup,1,z

/auto

!set,last

/title,pdrcoils1%rn% I(TF)=%itf%A, LC=%time_pt%

esel,s,mat,,2

nsle

plns,volt

plns,b,sum

plns,s,eqv

plns,u,z

plns,u,sum

esel,s,mat,,1,5

esel,r,type,,1,3,2

nsle

plns,volt

esel,s,type,,1,4,3

nsle

plns,u,sum

esel,s,mat,,10,11

nsle

plns,s,eqv

esel,s,mat,,2

nsle

csys,1

nsel,r,loc,x,,r0

nsel,r,loc,z,-.25,.25

esln,,1

/view

/type,1,0

/title,pdrcoils1%rn% I(TF)=%itf%A, LC=%time_pt%, TF Inboard Legs (+/-0.25 m) F=JxBdV

PLVECT,FMAG, , , ,VECT,ELEM,ON,0

/title,pdrcoils1%rn% I(TF)=%itf%A, LC=%time_pt%, TF Inboard Legs (+/-0.25 m) von Mises

/view,1,1,,1

/type

plns,s,eqv

/title,pdrcoils1%rn% I(TF)=%itf%A, LC=%time_pt%, CS von Mises

cmsel,s,cs_elem

nsle

rsys,1

plns,s,eqv

plns,s,z

plns,s,y

rsys

/com Sum forces on Ring coils

esel,s,mat,,10

nsle

csys,1

*get,thmn,node,,mnloc,y

*get,thmx,node,,mxloc,y

zsym=nint(360/(thmx-thmn))

*do,j,3,6

rsys,1

cmsel,s,pf%j%_elem

etab,fz,f,z

cmsel,s,pf%j%u_elem

ssum

*get,fz%j%u,ssum,,item,fz

fu=nint(zsym*fz%j%u/1000)

cmsel,s,pf%j%l_elem

ssum

*get,fz%j%l,ssum,,item,fz

fl=nint(zsym*fz%j%l/1000)

cmsel,s,pf%j%_elem

esel,u,mat,,9

nsle

/type,1,0

/title,pdrcoils1%rn% I(TF)=%itf%A, LC=%time_pt%, Fz(PF%j%U/L)=%fu%/%fl% kN

PLVECT,FMAG, , , ,VECT,ELEM,ON,0

/type

/title,pdrcoils1%rn% I(TF)=%itf%A, LC=%time_pt%, PF%j% von Mises, Fz(U/L)=%fu%/%fl%kN

plns,s,eqv

*enddo

/title,pdrcoils1%rn% I(TF)=%itf%A, LC=%time_pt%, Force Density (N/mm**3)

/com Force Density

/triad,off

esel,s,type,,1,3,2

esel,u,mat,,9

nsle

etab,fmag,fmag,sum

etab,volu,volu

SEXP,fd,FMAG,VOLU,1,-1, 

sadd,mfd,fd,,1e-6

/view,1,-1,1,.25

esel,s,mat,,1

esel,r,type,,1,3,2

nsle

plet,mfd,avg

esel,s,mat,,2

nsle

/view,1,1,0.4,.25

plet,mfd,avg

esel,s,mat,,3,5

nsle

/view,1,1

plet,mfd,avg

allsel

/title,pdrcoils1%rn% I(TF)=%itf%A, LC=%time_pt%

/type

fini

/exit,all

!/exit,nosa

/eof
4.2 Effects of Orthotropic and Isotropic Resistivity Values

There is a temptation to model the TF winding packs with orthotropic resistivities. After all, the resistance of the coil pack is small in the winding direction (as determined by the conductor material), and large in the two orthogonal directions (as determined by turn and layer insulation). However, subtitles in the local element-based coordinate system can produce non-physical results when this approach is used. Fig. 4.2-1 shows plots of the current density distribution in winding packs with orthotropic (upper plot) and isotropic (lower plot) coil pack resistivities. The current density distribution should be fairly uniform, with somewhat higher values in the inboard vertical legs where the cross-sectional area is lower than elsewhere in the winding.

The upper plot (with orthotropic resistivity) shows some variations which are clearly non-physical, particularly around the equatorial plane of the outboard leg. The lower plot (with isotropic resistivity) shows a much more uniform current density, with variations which are attributable to the reasons described above.  

Fig. 4.2-1 Current Density Distribution with Orthotropic (above) and Isotropic (below) Resistivity

